

**[CSE301 / Lecture 6]**  
**Zippers and derivatives of data types**

Noam Zeilberger

Ecole Polytechnique

9 October 2024

## Motivation

Many efficient algorithms use pointers to navigate within a data structure and perform destructive updates, e.g.:

```
# ... removing node w/two children from a BST
parent, current = node, node.right
while current.left is not None:
    parent, current = current, current.left
if parent is node:
    parent.right = current.right
else:
    parent.left = current.right
node.value = current.value
```

Can we perform such manipulations in a purely functional way?

## What is a “zipper”?<sup>1</sup>

A **zipper** is a pair of a *value* and a *one-hole context*.

Idea: represent a pointer by the value being pointed to, together with a representation of its surrounding context.

There is some beautiful theory behind this idea, which is also very useful in practice (e.g., used in the XMonad window manager!)

---

<sup>1</sup>The terminology comes from a 1997 article by Gérard Huet in the *Journal of Functional Programming*. “Going up and down in the structure is analogous to closing and opening a zipper in a piece of clothing, whence the name.”

## Example #1: a functional line editor using list zippers

How can we represent a “cursor” into a line of text, in a way that supports both efficient navigation and editing?

## Example #1: a functional line editor using list zippers

To make this question concrete, let's introduce a class of cursor types with some expected operations.

**class** *Cursor* *cur* **where**

```
cursor :: [a] → cur a           -- create a cursor to head of list
value :: cur a → a             -- return value at cursor
list :: cur a → [a]           -- return the underlying list
movL :: cur a → cur a         -- move the cursor to the left
movR :: cur a → cur a         -- move the cursor to the right
ovw :: a → cur a → cur a      -- overwrite item with new value
bks :: cur a → cur a         -- remove the item to the left
ins :: a → cur a → cur a      -- insert a new item to the left
```

## Example #1: a functional line editor using list zippers

Naive solution: store the index of the element in focus.

```
newtype NaiveCursor a = NC ([a], Int)
```

```
instance Cursor NaiveCursor where
```

```
  cursor xs = NC (xs, 0)
```

```
  value (NC (xs, i)) = xs !! i
```

```
  list (NC (xs, _)) = xs
```

```
  movL (NC (xs, i)) = NC (xs, i - 1)
```

```
  movR (NC (xs, i)) = NC (xs, i + 1)
```

```
  ovw y (NC (xs, i)) = NC (take i xs ++ [y] ++ drop (i + 1) xs, i)
```

```
  bks (NC (xs, i)) =
```

```
    NC (take (i - 1) xs ++ [xs !! i] ++ drop (i + 1) xs, i - 1)
```

```
  ins y (NC (xs, i)) =
```

```
    NC (take i xs ++ [y, xs !! i] ++ drop (i + 1) xs, i + 1)
```

Problem: *value*, *ovw*, *bks*, and *ins* all take time  $O(n)$ !

## Example #1: a functional line editor using list zippers

Better solution: use a zipper!

```
newtype ListZip a = LZ ([a], a, [a])  
instance Cursor ListZip where  
  cursor xs = LZ ([], head xs, tail xs)  
  value (LZ (ls, x, rs)) = x  
  list (LZ (ls, x, rs)) = reverse ls ++ [x] ++ rs  
  movL (LZ (l : ls, x, rs)) = LZ (ls, l, x : rs)  
  movR (LZ (ls, x, r : rs)) = LZ (x : ls, r, rs)  
  ovw y (LZ (ls, x, rs)) = LZ (ls, y, rs)  
  bks (LZ (_ : ls, x, rs)) = LZ (ls, x, rs)  
  ins y (LZ (ls, x, rs)) = LZ (y : ls, x, rs)
```

Now almost all operations are  $O(1)$ .

## Example #1: a functional line editor using list zippers

In this example, a one-hole context is just a pair of lists: the list of values to the left and the list of values to the right.

(But observe that the values to the left are stored in reverse order, i.e., in order of distance from the hole.)

[Let's load the file `ListZipper.hs` to see this in action...]

## Example #2: binary tree zippers

Recall the data type of binary trees with labelled leaves:

```
data Bin a = L a | B (Bin a) (Bin a)
```

Now a data type of one-hole contexts for binary trees:

```
data BinCxt a = Hole  
  | B0 (BinCxt a) (Bin a)  
  | B1 (Bin a) (BinCxt a)
```

Idea: a value of type *BinCxt a* provides a description of what we see on a path from the hole to the root of the tree. (Observe that *BinCxt a* is isomorphic to a list of pairs [(*Bool*, *Bin a*)].)

## Example #2: binary tree zippers

This path description is interpreted by the “plugging” function:

$$\begin{aligned} \text{plug} &:: \text{BinCxt } a \rightarrow \text{Bin } a \rightarrow \text{Bin } a \\ \text{plug } \text{Hole } t &= t \\ \text{plug } (\text{B0 } c \ t2) \ t &= \text{plug } c \ (\text{B } t \ t2) \\ \text{plug } (\text{B1 } t1 \ c) \ t &= \text{plug } c \ (\text{B } t1 \ t) \end{aligned}$$

A (binary tree) zipper is a pair of a one-hole context and a tree:

$$\text{type BinZip } a = (\text{BinCxt } a, \text{Bin } a)$$

Intuitively, we can think of a zipper  $(c, t)$  as defining a (purely functional) pointer to a subtree of  $u = \text{plug } c \ t$ .



## Example #2: binary tree zippers

Easy to implement operations for navigating through a tree:<sup>2</sup>

```
go_left :: BinZip a → Maybe (BinZip a)
go_left (c, B t1 t2) = Just (B0 c t2, t1)
go_left (c, L _)    = Nothing

go_right :: BinZip a → Maybe (BinZip a)
go_right (c, B t1 t2) = Just (B1 t1 c, t2)
go_right (c, L _)    = Nothing

go_down :: BinZip a → Maybe (BinZip a)
go_down (B0 c t2, t) = Just (c, B t t2)
go_down (B1 t1 c, t) = Just (c, B t1 t)
go_down (Hole, t)    = Nothing
```

---

<sup>2</sup>These operations can fail if one tries to navigate off the end of the tree. We chose to represent failure explicitly by using *Maybe* for the return type, but an alternative is to raise an exception.

## Example #2: binary tree zippers

Similarly easy to implement operations for performing local edits, such as say grafting another tree off to the left or right of the subtree in focus.

$$\begin{aligned} \text{graft\_left}, \text{graft\_right} &:: \text{Bin } a \rightarrow \text{BinZip } a \rightarrow \text{BinZip } a \\ \text{graft\_left } g (c, t) &= (c, B g t) \\ \text{graft\_right } g (c, t) &= (c, B t g) \end{aligned}$$

[Let's have a look at the solution set for Lab 4 to see how this can be used to solve the optional problem on random binary trees!]

## One-hole contexts and derivatives

There is a remarkable link between computing the types of zippers and *differential calculus*, summarized by the slogan that “the derivative of a type is its type of one-hole contexts.”<sup>3</sup>

More precisely, the type of one-hole contexts for a parameterized type  $T(a)$  may be computed as a partial derivative  $\frac{d}{da} T(a)$ .

The type of zippers is then obtained by multiplying  $a \cdot \frac{d}{da} T(a)$ .

---

<sup>3</sup>Which is also (almost) the title of an unpublished but very influential paper by Conor McBride, <http://strictlypositive.org/diff.pdf>

## Types as algebraic expressions

Recall the link we discussed in Lecture 1:

**data**  $\text{Either } a \ b = \text{Left } a \mid \text{Right } b \iff a + b$

**type**  $\text{Pair } a \ b = (a, b) \iff a \cdot b$

**data**  $\text{Void} \iff 0$

**data**  $() = () \iff 1$

**data**  $\text{Bool} = \text{True} \mid \text{False} \iff 2$

**data**  $\text{Maybe } a = \text{Nothing} \mid \text{Just } a \iff 1 + a$

Many algebraic laws are realized as type isomorphisms  
(e.g.,  $a + b = b + a \iff \text{Either } a \ b \cong \text{Either } b \ a$ ).

What about the laws of differentiation?

## Some laws of differentiation

$$\frac{d}{da}(S(a) + T(a)) = \frac{d}{da}S(a) + \frac{d}{da}T(a)$$

$$\frac{d}{da}(S(a) \cdot T(a)) = \frac{d}{da}S(a) \cdot T(a) + S(a) \cdot \frac{d}{da}T(a)$$

$$\frac{d}{da}(S(T(a))) = \frac{d}{db}S(b)|_{b=T(a)} \cdot \frac{d}{da}T(a)$$

These all have interpretations as poking holes in data types!

## Derivatives of data types: example

Consider the type of pairs of values of the same type:

**type** *Square*  $a = (a, a)$

A one-hole context for *Square*  $a$  is of the form  $(-, a)$  or  $(a, -)$ .

We can represent this with the type

**data** *SquareCxt*  $a = L\ a \mid R\ a$

which is isomorphic to the product type  $(Bool, a)$ .

Compare this with  $\frac{d}{da} a^2 = 2a$ .

## Derivatives of data types: example

Consider the type of lists:

**data** *List* *a* = *Nil* | *Cons* *a* (*List* *a*)

Now expand it algebraically, and differentiate:

$$\begin{aligned}L(a) &= 1 + a \cdot L(a) \\ &= 1 + a \cdot (1 + a \cdot L(a)) \\ &= 1 + a + a^2 + a^3 + \dots \\ &= (1 - a)^{-1}\end{aligned}$$

$$\frac{d}{da}L(a) = (1 - a)^{-2} = L(a)^2$$

We recover that *ListCxt* *a*  $\cong$  (*[a]*, *[a]*) and *ListZip* *a*  $\cong$  (*[a]*, *a*, *[a]*)!

## Derivatives of data types: example

Consider the type of binary trees:

**data**  $Bin\ a = L\ a \mid B\ (Bin\ a)\ (Bin\ a)$

Now expand it algebraically, and differentiate:

$$\begin{aligned}T(a) &= a + T(a)^2 \\ \frac{d}{da} T(a) &= 1 + 2T(a) \frac{d}{da} T(a) \\ &= \frac{1}{1 - 2T(a)}\end{aligned}$$

We recover that  $BinCxt\ a \cong List\ (Bool, Bin\ a)!$

## Some references

On zippers and derivatives of data types:

- The Haskell Wikibook on Zippers
- Gérard Huet, “The Zipper”, *JFP* 7:5, 1997
- Conor McBride, “The derivative of a regular type is its type of one-hole contexts”, unpublished manuscript, 2000

On the closely related theory of “combinatorial species”:

- André Joyal, “Une théorie combinatoire des séries formelles”, *Advances in Mathematics* 42(1), 1981
- Brent Yorgey, “Species and functors and types, oh my!”, in proceedings of *Haskell '10*, 2010 (see also his PhD thesis)